/*
 *  Licensed to the Apache Software Foundation (ASF) under one
 *  or more contributor license agreements.  See the NOTICE file
 *  distributed with this work for additional information
 *  regarding copyright ownership.  The ASF licenses this file
 *  to you under the Apache License, Version 2.0 (the
 *  "License"); you may not use this file except in compliance
 *  with the License.  You may obtain a copy of the License at
 *
 *    https://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing,
 *  software distributed under the License is distributed on an
 *  "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY
 *  KIND, either express or implied.  See the License for the
 *  specific language governing permissions and limitations
 *  under the License.
 */

plugins {
    id 'groovy'
    id 'java-library'
    id 'org.apache.grails.buildsrc.publish'
    id 'org.apache.grails.buildsrc.sbom'
}

version = projectVersion
group = 'org.apache.grails.data'

ext {
    gormApiDocs = true
    pomTitle = 'GORM for MongoDB'
    pomDescription = 'Provides a GORM Object Mapping implementations for the MongoDB Document Database'
    pomDevelopers =  [
            'graemerocher': 'Graeme Rocher',
            'jeffscottbrown': 'Jeff Brown',
            'burtbeckwith': 'Burt Beckwith'
    ]
}

dependencies {

    implementation platform(project(':grails-bom'))

    api 'org.mongodb:bson', {
        // api: AbstractBsonReader, AbstractBsonWriter, BsonBinary, BsonContextType, BsonDbPointer, BsonReader,
        //      BsonRegularExpression, BsonRenderMark, BsonTimestamp, BsonType, CodecRegistry, Decimal128, Document,
        //      EncoderContext, JsonWriterSettings, ObjectId
        // impl: JsonParseException
    }
    api project(':grails-datastore-core'), {
        // api: Criterion, Embedded, In, Junction, Like, PersistentEntity, Projection, PropertyCriterion,
        //      PropertyNameCriterion, Query, Session
        // impl: Custom, CustomTypeMarshaller, EntityReflector, MappingContext, MappingUtils, PersistentProperty,
        //       Property, ProxyHandler, ToOne
        if (excludeUnusedTransDeps) {
            // API dependencies in grails-core
            exclude group: 'jakarta.annotation', module: 'jakarta.annotation-api'
            exclude group: 'jakarta.inject', module: 'jakarta.inject-api'
            //exclude group: 'jakarta.persistence', module: 'jakarta.persistence-api' // Entity
            exclude group: 'org.apache.grails.bootstrap', module: 'grails-bootstrap'
            exclude group: 'org.apache.grails.data', module: 'grails-datastore-core'
            exclude group: 'org.apache.grails', module: 'grails-spring'
            exclude group: 'org.springframework', module: 'spring-beans'
            exclude group: 'org.springframework', module: 'spring-context'
            //exclude group: 'org.springframework', module: 'spring-core' // ConversionService
            exclude group: 'org.springframework', module: 'spring-tx'
            exclude group: 'org.springframework.boot', module: 'spring-boot'
            exclude group: 'org.springframework.boot', module: 'spring-boot-autoconfigure'
        }
    }
    api project(':grails-datamapping-core'), {
        // api: DetachedCriteria
        if (excludeUnusedTransDeps) {
            // API dependencies in grails-datamapping-core
            exclude group: 'jakarta.annotation', module: 'jakarta.annotation-api'
            //exclude group: 'jakarta.persistence', module: 'jakarta.persistence-api' // Entity
            exclude group: 'org.apache.grails.data', module: 'grails-datastore-core'
            exclude group: 'org.apache.grails.data', module: 'grails-datamapping-validation'
            //exclude group: 'org.springframework', module: 'spring-context' // Validator
            exclude group: 'org.springframework', module: 'spring-jdbc'
            exclude group: 'org.springframework', module: 'spring-tx'
        }
    }

    implementation 'org.springframework:spring-tx', {
        // impl: InvalidDataAccessResourceUsageException
    }
    implementation 'org.apache.groovy:groovy', {
        // impl: GStringImpl, NullObject
        // comp: @CompileStatic, @CompileDynamic
    }

    testImplementation 'org.apache.groovy:groovy-json', {
        // impl: JsonSlurper
    }
    testImplementation 'org.spockframework:spock-core'
    testImplementation project(':grails-testing-support-mongodb')
}

apply {
    from rootProject.layout.projectDirectory.file('gradle/java-config.gradle')
    from rootProject.layout.projectDirectory.file('gradle/code-style-config.gradle')
    from rootProject.layout.projectDirectory.file('gradle/mongodb-test-config.gradle')
    from rootProject.layout.projectDirectory.file('gradle/docs-config.gradle')
}